

Programming Languages: Functional Programming

6. Program Calculation: Work Less by Promising More

Shin-Cheng Mu

Autumn 2025

Correct by Construction

Dijkstra: “The only effective way to raise the confidence level of a program significantly is to give a convincing proof of its correctness. But one should not first make the program and then prove its correctness, because then the requirement of providing the proof would only increase the poor programmer’s burden. On the contrary: the programmer should ...”

“...[let] correctness proof and program grow hand in hand: with the choice of the structure of the correctness proof one designs a program for which this proof is applicable.”

- A *steep list* is a list in which every element is larger than the sum of those to its right:

$$\begin{aligned} \text{steep} &:: \text{List Int} \rightarrow \text{Bool} \\ \text{steep } [] &= \text{True} \\ \text{steep } (x : xs) &= \text{steep } xs \wedge x > \text{sum } xs. \end{aligned}$$

- The definition above, if executed directly, is an $O(n^2)$ program. Can we do better?
- Just now we learned to construct a generalised function which takes more input. This time, we try the dual technique: to construct a function returning more results.

Deriving Programs from Specifications

- In functional program derivation, the specification itself is a function, albeit probably not an efficient one.
- From the specification we construct a function that equals the specification.
- The calculation is the proof.
- In the previous class to proceed by expanding and reducing the definitions, until we obtain an inductive definition of the specification.
- But that does not work all the time.
- In this lecture we review some techniques that might work for more cases.

Generalise by Returning More

- Recall that $\text{fst } (a, b) = a$ and $\text{snd } (a, b) = b$.
- It is hard to quickly compute *steep* alone. But if we define

$$\text{steepsum } xs = (\text{steep } xs, \text{sum } xs),$$

- and manage to synthesise a quick definition of *steepsum*, we can implement *steep* by $\text{steep} = \text{fst} \cdot \text{steepsum}$.
- We again proceed by case analysis. Trivially,

$$\text{steepsum } [] = (\text{True}, 0).$$

1 Tupling

Steep Lists

Deriving for the Non-Empty Case

For the case for non-empty inputs:

```

steepsum (x : xs)
= { definition of steepsum }
  (steep (x : xs), sum (x : xs))
= { definitions of steep and sum }
  (steep xs ∧ x > sum xs, x + sum xs)
= { extracting sub-expressions involving xs }
  let (b, y) = (steep xs, sum xs)
  in (b ∧ x > y, x + y)
= { definition of steepsum }
  let (b, y) = steepsum xs
  in (b ∧ x > y, x + y).

```

Synthesised Program

- We have thus come up with a $O(n)$ time program:

```

steep          = fst · steepsum
steepsum []    = (True, 0)
steepsum (x : xs) = let (b, y) = steepsum xs
                    in (b ∧ x > y, x + y),

```

- Again we observe the phenomena that a more general function is easier to implement.

2 Accumulating Parameters

Reversing a List

- The function *reverse* is defined by:

```

reverse []      = [],
reverse (x : xs) = reverse xs ++ [x].

```

- E.g. $reverse [1, 2, 3, 4] = ((([] ++ [4]) ++ [3]) ++ [2]) ++ [1] = [4, 3, 2, 1]$.
- But how about its time complexity? Since $(++)$ is $O(n)$, it takes $O(n^2)$ time to revert a list this way.
- Can we make it faster?

2.1 Fast List Reversal

Introducing an Accumulating Parameter

- Let us consider a generalisation of *reverse*. Define:

```

revcat      :: [a] → [a] → [a]
revcat xs ys = reverse xs ++ ys.

```

- If we can construct a fast implementation of *revcat*, we can implement *reverse* by:

```
reverse xs = revcat xs [].
```

Reversing a List, Base Case

Let us use our old trick. Consider the case when *xs* is []:

```

revcat [] ys
= { definition of revcat }
  reverse [] ++ ys
= { definition of reverse }
  [] ++ ys
= { definition of (++) }
  ys.

```

Reversing a List, Inductive Case

Case $x : xs$:

```

revcat (x : xs) ys
= { definition of revcat }
  reverse (x : xs) ++ ys
= { definition of reverse }
  (reverse xs ++ [x]) ++ ys
= { since (xs ++ ys) ++ zs = xs ++ (ys ++ zs) }
  reverse xs ++ ([x] ++ ys)
= { definition of revcat }
  revcat xs (x : ys).

```

Linear-Time List Reversal

- We have therefore constructed an implementation of *revcat* which runs in linear time!

```

revcat [] ys      = ys
revcat (x : xs) ys = revcat xs (x : ys).

```

- A generalisation of *reverse* is easier to implement than *reverse* itself? How come?
- If you try to understand *revcat* operationally, it is not difficult to see how it works.

- The partially reverted list is *accumulated* in *ys*.
- The initial value of *ys* is set by $reverse\ xs = revcat\ xs\ []$.
- Hmm... it is like a *loop*, isn't it?

2.2 Tail Recursion and Loops

Tracing Reverse

```

reverse [1, 2, 3, 4]
= revcat [1, 2, 3, 4] []
= revcat [2, 3, 4] [1]
= revcat [3, 4] [2, 1]
= revcat [4] [3, 2, 1]
= revcat [] [4, 3, 2, 1]
= [4, 3, 2, 1]

reverse xs      = revcat xs []
revcat [] ys    = ys
revcat (x : xs) ys = revcat xs (x : ys)

xs, ys ← XS, [];
while xs ≠ [] do
  xs, ys ← (tail xs), (head xs : ys);
return ys

```

Tail Recursion

- Tail recursion: a special case of recursion in which the last operation is the recursive call.

$$\begin{aligned}
 f\ x_1 \dots x_n &= \{\text{base case}\} \\
 f\ x_1 \dots x_n &= f\ x'_1 \dots x'_n
 \end{aligned}$$

- To implement general recursion, we need to keep a stack of return addresses. For tail recursion, we do not need such a stack.
- Tail recursive definitions are like loops. Each x_i is updated to x'_i in the next iteration of the loop.
- The first call to f sets up the initial values of each x_i .

Accumulating Parameters

- To efficiently perform a computation (e.g. *reverse xs*), we introduce a generalisation with an extra parameter, e.g.:

$$revcat\ xs\ ys = reverse\ xs\ ++\ ys.$$

- Try to derive an efficient implementation of the generalised function. The extra parameter is usually used to “accumulate” some results, hence the name.
 - To make the accumulation work, we usually need some kind of associativity.
- A technique useful for, but not limited to, constructing tail-recursive definition of functions.

Accumulating Parameter: Another Example

- Recall the “sum of squares” problem:

$$\begin{aligned}
 sumsq\ [] &= 0 \\
 sumsq\ (x : xs) &= square\ x + sumsq\ xs.
 \end{aligned}$$

- The program still takes linear space (for the stack of return addresses). Let us construct a tail recursive auxiliary function.
- Introduce $ssp\ xs\ n = sumsq\ xs + n$.
- Initialisation: $sumsq\ xs = ssp\ xs\ 0$.
- Construct ssp :

$$\begin{aligned}
 ssp\ []\ n &= 0 + n = n \\
 ssp\ (x : xs)\ n &= (square\ x + sumsq\ xs) + n \\
 &= sumsq\ xs + (square\ x + n) \\
 &= ssp\ xs\ (square\ x + n).
 \end{aligned}$$

2.3 Being Quicker by Doing More!

Being Quicker by Doing More?

- A more generalised program can be implemented more efficiently?
 - A common phenomena! Sometimes the less general function cannot be implemented inductively at all!
 - It also often happens that a theorem needs to be generalised to be proved. We will see that later.
- An obvious question: how do we know what generalisation to pick?
- There is no easy answer — finding the right generalisation one of the most difficulty act in programming!
- For the past few examples, we choose the generalisation to exploit associativity.
- Sometimes we simply generalise by examining the form of the formula.

Labelling a List

- Consider the task of labelling elements in a list with its index.

$$\begin{aligned}
 index &:: List\ a \rightarrow List\ (Int, a) \\
 index &= zip\ [0..]
 \end{aligned}$$

- To construct an inductive definition, the case for $[]$ is easy. For the $x : xs$ case:

$$\begin{aligned} & \text{index } (x : xs) \\ &= \text{zip } [0..] (x : xs) \\ &= (0, x) : \text{zip } [1..] xs \end{aligned}$$

- Alas, $\text{zip } [1..]$ cannot be fold back to index !
- What if we turn the varying part into...a variable?

- For the case $x : xs$ we reason:

$$\begin{aligned} & \text{sum } (\text{revcat } (x : xs) \text{ } ys) \\ &= \text{sum } (\text{revcat } xs (x : ys)) \\ &= \{ \text{induction hypothesis} \} \\ & \quad \text{sum } xs + \text{sum } (x : ys) \\ &= \text{sum } xs + x + \text{sum } ys \\ &= \text{sum } (x : xs) + \text{sum } ys \end{aligned}$$

Work Less by Proving More

Labelling a List, Second Attempt

- Introduce $\text{idxFrom} :: \text{List } a \rightarrow \text{Int} \rightarrow \text{List } (Int, a)$:

$$\text{idxFrom } xs \ n = \text{zip } [n..] \ xs$$

- Initialisation: $\text{index } xs = \text{idxFrom } xs \ 0$.
- We reason:

$$\begin{aligned} & \text{idxFrom } (x : xs) \ n \\ &= \text{zip } [n..] (x : xs) \\ &= (n, x) : \text{zip } [n + 1..] xs \\ &= (n, x) : \text{idxFrom } xs \ (n + 1) \end{aligned}$$

- A stronger theorem is easier to prove! Why is that?

- By strengthening the theorem, we also have a stronger induction hypothesis, which makes an inductive proof possible.

- Finding the right generalisation is an art — it's got to be strong enough to help the proof, yet not too strong to be provable.

- The same with programming. By generalising a function with additional arguments, it is passed more information it may use, thus making an inductive definition possible.

- The speeding up of revcat , in retrospect, is an accidental “side effect” — revcat , being inductive, goes through the list only once, and is therefore quicker.

3 Proof by Strengthening

Summing Up a List in Reverse

- Prove: $\text{sum} \cdot \text{reverse} = \text{sum}$, using the definition $\text{reverse } xs = \text{revcat } xs \ []$. That is, proving $\text{sum } (\text{revcat } xs \ []) = \text{sum } xs$.
- Base case trivial. For the case $x : xs$:

$$\begin{aligned} & \text{sum } (\text{reverse } (x : xs)) \\ &= \text{sum } (\text{revcat } (x : xs) \ []) \\ &= \text{sum } (\text{revcat } xs \ [x]) \end{aligned}$$

- Then we are stuck, since we cannot use the induction hypothesis $\text{sum } (\text{revcat } xs \ []) = \text{sum } xs$.
- Again, generalise $[x]$ to a variable.

A Real Case

- A property I actually had to prove for a paper:

$$\begin{aligned} & \text{msmp } (\text{trim } (x : xs)) = \text{msmp } (\text{trim } (x : \text{win } xs)) \\ & \Leftarrow \text{msmp } (\text{trim } (x : xs)) >_d \text{mds } xs \end{aligned}$$

- It took me a week to construct the right generalisation:

$$\begin{aligned} & \text{msmp } (\text{trim } (zs ++ xs)) = \text{msmp } (\text{trim } (zs ++ \text{win } xs)) \\ & \Leftarrow \text{msmp } (\text{trim } (zs ++ xs)) >_d \text{mds } xs \end{aligned}$$

Summing Up a List in Reverse, Second Attempt

- Second attempt: prove a lemma:

$$\text{sum } (\text{revcat } xs \ ys) = \text{sum } xs + \text{sum } ys$$

- By letting $ys = []$ we get the previous property.

- Once the right property is found, the actual proof was done in about 20 minutes.

- “Someone once described research as ‘finding out something to find out, then finding it out’; the first part is often harder than the second.”

Remark

- The *sum · reverse* example is superficial — the same property is much easier to prove using the $O(n^2)$ -time definition of *reverse*.
- That's one of the reason we defer the discussion about efficiency — to prove properties of a function we sometimes prefer to roll back to a slower version.
- In our exercises there is an example where you need *revcat* to prove a property about *reverse*.
 - Show that $\text{reverse} \cdot \text{reverse} = \text{id}$